

A large, vibrant background graphic featuring a repeating pattern of colorful pinwheel-like shapes. Each pinwheel is composed of several triangular segments in blue, yellow, red, and green, radiating from a central white point. The overall effect is a dynamic, swirling pattern.

Efeitos gráficos com o LibreLogo

Por Gilvan Vilarim

LibreLogo é um ambiente de programação usado para a criação de desenhos na tela de documento do Writer, o processador de textos do LibreOffice.


Com a barra de ferramentas Logo, é possível construir desenhos que se transformam em gráficos vetoriais, ou seja, objetos gráficos do mesmo tipo daqueles criados com a barra de Desenho.

Uma tartaruguinha funciona como um cursor de onde saem os desenhos. Uma vez prontos, basta selecionar e aproveitar o resultado em outros documentos.

Na edição 4 da LibreOffice Magazine, eu escrevi um artigo, para iniciantes, sobre o LibreLogo. Vale a pena ler!

Uma das aplicações possíveis do LibreLogo é o design gráfico. Podemos usar a tartaruga para criar diversos efeitos na folha de documento, principalmente efeitos geométricos. Para desenhos mais elaborados, é preciso criar **programas**, usando a linguagem do LibreLogo. Um programa em LibreLogo é um texto contendo diversos comandos, colocado dentro de um documento vazio do Writer.

COMO FAZER | *tutorial*



Cada documento deve ter apenas um programa. Para outro programa, crie outro arquivo vazio. Neste artigo mostraremos alguns programas prontos que podem ser facilmente adaptados para gerar outros efeitos. Basta digitar o programa exatamente como está (ou Copiar e Colar) dentro de um documento inicialmente vazio.

Colcha de bolotas

O programa a seguir vai encher a folha com diversas bolotas coloridas. As cores são escolhidas aleatoriamente a cada execução.

Para executar, na barra Logo, clique no botão de Play (com a figura da seta verde).

Depois é só aguardar a tartaruga terminar!

O programa a seguir vai encher a folha com diversas bolotas coloridas.

As cores são escolhidas aleatoriamente a cada execução.

```
; Um programa para desenhar colcha de bolotas
TARTARUGA
APARECERTAT
PARACENTRO
USARNADA

largura    = TAMPÁG[0]
altura     = TAMPÁG[1]
tambola    = largura / 9
qthorizontal = INT largura / tambola
qtvertical  = INT altura / tambola

PARA lin EM INTERVALO qtvertical [
  PARA col EM INTERVALO qthorizontal [
    POSICIONAR [tambola / 2 + col * tambola, tambola / 2 + lin *
tambola]
    MUDARCORDAPINTURA [4 + ALEATÓRIO 20]
    CÍRCULO tambola
  ]
]
; Fim do programa
```

COMO FAZER | *tutorial*



Legal, não?

- Você pode alterar a quantidade de bolotas alterando a linha do programa que contém o cálculo $\text{largura} / 9$ (o 9 é a quantidade de bolotas em uma linha da folha; experimente colocar outro valor, como 4 ou 20).

O programa sabe desenhar as bolotas em função do tamanho da folha de papel.

- Caso as bolotas fiquem sobrepostas ao seu programa, insira uma quebra de página no seu início e deixe o programa ficar na segunda folha (o desenho sempre é feito na primeira folha do documento).

A Figura 1 mostra um exemplo de resultado.



Figura 1 – Colcha de bolotas



Uma mandala

O programa a seguir é pequeno, mas faz mandalas bem bonitas.

```
; Um programa para desenhar uma mandala
APRENDER poligono n
  REPETIR n [
    PARAFRENTE n*10
    PARADIREITA 360 / n
  ]
FIM

TARTARUGA
PARACENTRO
MUDARESPESURADOLÁPIS 3
REPETIR 24 [
  MUDARCORDOLÁPIS “~TURQUESA”
  poligono 7
  PARADIREITA 15
]
; Fim do programa
```

Você pode mudar o programa de várias maneiras.

- Tente, por exemplo, alterar o valor na linha **MUDARESPESURADOLÁPIS**. Quanto maior esse valor, mais grossa ficará a figura.
- A cor também pode ser mudada na linha **MUDARCORDOLÁPIS**. Basta trocar a palavra **TURQUESA** (o til é para o LibreLogo criar variações aleatórias nessa cor).

- Alguns exemplos de cores são **VIOLETA**, **OURO**, **LIMA**, **VERMELHO**, etc (há uma lista completa na Ajuda do LibreLogo).
- A base da mandala é desenhar vários polígonos de 7 lados. Veja que é fácil mudar isso para outra quantidade de lados do polígono (mude a linha que indica o valor 7).

A Figura 2 mostra um exemplo.



Figura 2 – Mandala

Bordas na página

O programa a seguir desenha uma borda em torno da folha de documento.

```
; Um programa para fazer uma borda em volta da folha  
TARTARUGA  
PARACENTRO
```

```
ms = 1cm  
mi = 2cm  
me = 3cm  
md = 4cm  
zerox = me  
zeroy = ms
```

```
fimx = TAMANHODAPÁGINA[0] - md  
fimy = TAMANHODAPÁGINA[1] - mi
```

```
USARNADA  
POSICIONAR [zerox, zeroy]
```

```
USARLÁPIS  
MUDARESPESURADOLÁPIS 6  
MUDARESTILODOLÁPIS "PONTILHADO"  
MUDARCORDOLÁPIS QUALQUER
```

```
POSICIONAR [fimx, zeroy]  
POSICIONAR [fimx, fimy]  
POSICIONAR [zerox, fimy]  
POSICIONAR [zerox, zeroy]  
; Fim do programa
```



- A cada execução, a borda sairá com uma cor diferente, por causa da palavra **QUALQUER**.
- Para mudar as margens utilizadas, altere as primeiras linhas contendo os valores em cm, que representam, respectivamente, as margens superior, inferior, esquerda e direita.
- Se você mudar a palavra **PONTILHADO** para **TRACEJADO**, obterá outro efeito ao desenhar. Veja um exemplo na Figura 3.




Figura 3 – Bordas na página

Raios coloridos

O programa a seguir não chega a ser uma maravilha da computação gráfica, mas cria um belo efeito de raios em toda a folha do documento.



```
;Desenho de raios coloridos
DESAPARECERAT
TARTARUGA
PARACENTRO
MUDARESPESSURADOLÁPIS 2
USARNADA

fimx = INT (TAMANHODAPÁGINA[0])
fimy = INT (TAMANHODAPÁGINA[1])
passox = INT (fimx / 10)
passoy = INT (fimy / 10)

MUDARCORDAPINTURA "PRETO"
MUDARCORDOLÁPIS "PRETO"
RETÂNGULO [fimx, fimy]

PARA ct EM INTERVALO 0 fimx passoX [
  USARNADA
  POSICIONAR [ct ,0]
  USARLÁPIS
  MUDARCORDOLÁPIS "~AZUL"
  POSICIONAR [fimx - ct ,fimy]
]

PARA ct EM INTERVALO 0 fimy passoY [
  USARNADA
  POSICIONAR [0,ct]
  USARLÁPIS
  MUDARCORDOLÁPIS "~AZUL"
  POSICIONAR [fimx, fimy - ct]
]

USARNADA
PARACENTRO
USARLÁPIS
MUDARCORDAPINTURA "PRETO"
MUDARTRANSPARÊNCIADAPINTURA 100

PARA ct EM INTERVALO passoX fimx passoX [
  MUDARCORDOLÁPIS "~VIOLETA"
  CÍRCULO ct
]
; Fim do programa
```



- O fundo da folha será preto, e o desenho das linhas será azul e violeta (mexa no programa para mudar essas cores).
- Ele também é capaz de acompanhar o tamanho da folha do documento.
- Para ter mais ou menos linhas na tela, mude os comandos do programa onde há `fimx / 10` ou `fimy / 10`.
- Se você colocar 5 em vez de 10, terá um espaçamento maior entre cada linha desenhada.

Veja como fica na Figura 4.




Figura 4 – Raios coloridos

Roda colorida

O programa a seguir desenha uma roda colorida, onde cada raio da roda é feito com linhas pontilhadas.



```
; Roda com arcos coloridos
APRENDER fazroda tamanho quant
ESCREVER "Uma roda colorida!"
REPETIR quant [
  MUDARCORDOLÁPIS QUALQUER
  PARAFRENTE tamanho
  PARATRÁS tamanho
  PARADIREITA 360/quant
]
FIM

TARTARUGA
PARACENTRO
DESAPARECERTAT
MUDARESPESURADOLÁPIS 5
MUDARESTILODOLÁPIS "PONTILHADO"

fazroda 200 40
; Fim do programa
```

Este programa tem um capricho: ele exibe uma mensagem em uma janelinha, indicado que vai começar a desenhar.

- A palavra **QUALQUER** também é uma indicação de que cada linha desenhada terá uma cor aleatória.
- O valor 200 é o raio da roda, medido em pontos.
- O valor 40 é a quantidade de raios. Experimente alterar os valores!

Veja ao lado a Figura 5.




Figura 5 – Roda colorida

COMO FAZER | *tutorial*

Indo além

Existem infinitas possibilidades de desenhos com o LibreLogo. Quanto mais você conhecer os comandos e os recursos de programação da linguagem, mais criatividade poderá usar nos seus efeitos. Se você já possui conhecimentos de programação, observe que o LibreLogo permite trabalhar com variáveis, repetições, cálculos, sub-rotinas, etc.

Agora é você quem vai dar asas à imaginação...



Gilvan Vilarim - Tradutor do LibreLogo para o português do Brasil. Bacharel em Informática pelo IME-UERJ, mestre em Engenharia de Sistemas e Computação pela COPPE-UFRJ, e doutor em Serviço Social pela ESS-UFRJ. Professor desde 1995, atuando principalmente no ensino técnico e de graduação na área de computação; trabalhou em instituições públicas e particulares ao longo dos anos. Atualmente é docente do Instituto Federal de Educação, Ciência e Tecnologia do Rio de Janeiro (IFRJ). Para mais informações, procure-o na Plataforma Lattes.